Chapter 21
Riemann Curvature Tensor

Catching a lion, the Einstein's method: Enter the cage and lock it from inside. Then the Universe will be subdivided into two disjoint regions in such a way that you are in one of them and the lion is in the other one. It depends on one's point of view whom to consider caught; for convenience, let's say it's the lion.

Suppose we have a coordinate system x^μ in a region of an n-dimensional Riemann (or pseudo-Riemann) manifold [20]. Components of the metric tensor $g_{\mu\nu}$ are given as functions of x^μ. We want to calculate the Riemann curvature tensor $R^\mu{}_{\nu\alpha\beta}$ and related quantities (the Ricci tensor $R_{\mu\nu}$, the scalar curvature R).

The metric tensor is symmetric; therefore, it is reasonable to ask the user to provide only the components with $\mu \geq \nu$. If the user gives an argument having a wrong shape, we print an error message and abort the calculation. We shall also need the contravariant metric tensor $g^{\mu\nu}$ defined by $g^{\mu\lambda} g_{\lambda\nu} = \delta^\mu_\nu$.

```
In[1] := Metric[g0_] := Module[{n = Length[g0], g, gu},
    Do[If[Length[g0[[μ]]]=!=μ, Message[Metric :: shape]; Abort[]], {μ, n}];
    g = Table[If[μ ≥ ν, g0[[μ, ν]], g0[[ν, μ]]], {μ, n}, {ν, n}];
    gu = Simplify[Inverse[g]]; {n, g, gu}]
In[2] := Metric :: shape = "Wrong shape of the argument";
```

Next we calculate the Christoffel symbols

$$\Gamma_{\lambda\mu\nu} = \frac{1}{2}\left(\partial_\mu g_{\lambda\nu} + \partial_\nu g_{\lambda\mu} - \partial_\lambda g_{\mu\nu}\right)$$

and $\Gamma^\lambda{}_{\mu\nu} = g^{\lambda\rho}\Gamma_{\rho\mu\nu}$. They are symmetric in μ and ν; therefore, we calculate them only at $\nu \leqslant \mu$ and reuse the calculated values at $\nu > \mu$. If the optional parameter PrintNonZero is True, nonzero components are printed (following the tradition, in the printed results all indices vary from 0 to $n-1$).

```
In[3] := Christoffel[{n_, g_, gu_}, OptionsPattern[]] := Module[{Γ, Γu},
  Γ = Γu = Table[0, {λ, n}, {μ, n}, {ν, n}];
  Do[Γ[[λ, μ, ν]] = Simplify[(D[g[[λ, ν]], x[μ]] + D[g[[λ, μ]], x[ν]]−
        D[g[[μ, ν]], x[λ]])/2];
```

A. Grozin, *Introduction to Mathematica® for Physicists*, Graduate Texts in Physics,
DOI 10.1007/978-3-319-00894-3_21,

```
      If[μ ≠ ν,Γ[[λ,ν,μ]] = Γ[[λ,μ,ν]]],
    {λ,n},{μ,n},{ν,μ}];
  Do[Γu[[λ,μ,ν]] = Simplify[Sum[gu[[λ,ρ]] * Γ[[ρ,μ,ν]],{ρ,n}]];
      If[μ ≠ ν,Γu[[λ,ν,μ]] = Γu[[λ,μ,ν]]],
    {λ,n},{μ,n},{ν,μ}];
  If[OptionValue[PrintNonZero],
    Do[If[Γ[[λ,μ,ν]]=!=0,Print["Γ",λ − 1,μ − 1,ν − 1," ",Γ[[λ,μ,ν]]]],
      {λ,n},{μ,n},{ν,μ}];
    Do[If[Γu[[λ,μ,ν]]=!=0,Print["Γu",λ − 1,μ − 1,ν − 1," ",Γu[[λ,μ,ν]]]],
      {λ,n},{μ,n},{ν,μ}]];
  {Γ,Γu}]
In[4] := Options[Christoffel] = {PrintNonZero → True};
```

Finally, we calculate the Riemann tensor

$$R_{\alpha\beta\mu\nu} = g_{\alpha\lambda}\left(\partial_\mu \Gamma^\lambda{}_{\beta\nu} - \partial_\nu \Gamma^\lambda{}_{\beta\mu}\right) + \Gamma_{\alpha\lambda\mu}\Gamma^\lambda{}_{\beta\nu} - \Gamma_{\alpha\lambda\nu}\Gamma^\lambda{}_{\beta\mu}\,,$$

the Ricci tensor $R_{\mu\nu} = g^{\alpha\beta}R_{\alpha\mu\beta\nu}$, and the scalar curvature $R = g^{\mu\nu}R_{\mu\nu}$. The Riemann tensor has the properties

$$R_{\alpha\beta\mu\nu} = -R_{\beta\alpha\mu\nu} = -R_{\alpha\beta\nu\mu} = R_{\mu\nu\alpha\beta}\,,$$

and we use them to avoid unnecessary calculations.

```
In[5] := Riemann[{n_,g_,gu_},OptionsPattern[]] := Module[{Γ,Γu,
    R = Table[0,{α,n},{β,n},{μ,n},{ν,n}],R2 = Table[0,{μ,n},{ν,n}],R0},
  {Γ,Γu} = Christoffel[{n,g,gu}];
  Do[R[[α,β,μ,ν]] = R[[β,α,ν,μ]] = Simplify[Sum[
        g[[α,λ]] * (D[Γu[[λ,β,ν]],x[μ]] − D[Γu[[λ,β,μ]],x[ν]])
        + Γ[[α,λ,μ]] * Γu[[λ,β,ν]] − Γ[[α,λ,ν]] * Γu[[λ,β,μ]],{λ,n}]];
      R[[β,α,μ,ν]] = R[[α,β,ν,μ]] = −R[[α,β,μ,ν]];
      If[μ ≠ α,R[[μ,ν,α,β]] = R[[ν,μ,β,α]] = R[[α,β,μ,ν]];
        R[[ν,μ,α,β]] = R[[μ,ν,β,α]] = −R[[α,β,μ,ν]]],
    {α,2,n},{β,α − 1},{μ,2,α},{ν,If[μ === α,β,μ − 1]}];
  Do[R2[[μ,ν]] = Simplify[Sum[gu[[α,β]] * R[[α,μ,β,ν]],{α,n},{β,n}]];
      If[μ ≠ ν,R2[[ν,μ]] = R2[[μ,ν]]],
    {μ,n},{ν,μ}];
  R0 = Simplify[Sum[gu[[μ,ν]] * R2[[μ,ν]] * If[μ ≠ ν,2,1],{μ,n},{ν,μ}]];
  If[OptionValue[PrintNonZero],
    Do[If[R[[α,β,μ,ν]]=!=0,
        Print[R,α − 1,β − 1,μ − 1,ν − 1," ",R[[α,β,μ,ν]]]],
      {α,2,n},{β,α − 1},{μ,2,α},{ν,If[μ === α,β,μ − 1]}];
    Do[If[R2[[μ,ν]]=!=0,Print["R",μ − 1,ν − 1," ",R2[[μ,ν]]]],{μ,n},{ν,μ}];
    If[R0=!=0,Print["R ",R0]]];
  {R,R2,R0}]
In[6] := Options[Riemann] = {PrintNonZero → True};
```

Let's consider an example: the Schwarzschild metric

$$ds^2 = \left(1 - \frac{r_0}{r}\right)dt^2 - \frac{dr^2}{1 - \frac{r_0}{r}} - r^2\left(d\theta^2 + \sin^2\theta\, d\varphi^2\right)\,.$$

First we give names to the coordinates.

In[7] := Evaluate[Table[$x[\mu]$, {μ,4}]] = {t, r, θ, φ};

Setting the Schwarzschild radius $r_0 = 1$, we obtain

**In[8] := Riemann[Metric[{{1 − 1/r}, {0, −1/(1 − 1/r)}, {0, 0, −r^2},
{0, 0, 0, −r^2 ∗ Sin[θ]^2}}]];**

$\Gamma 010\ \frac{1}{2r^2}$

$\Gamma 100\ -\frac{1}{2r^2}$

$\Gamma 111\ \frac{1}{2(-1+r)^2}$

$\Gamma 122\ r$

$\Gamma 133\ r\mathrm{Sin}[\theta]^2$

$\Gamma 221\ -r$

$\Gamma 233\ r^2\mathrm{Cos}[\theta]\mathrm{Sin}[\theta]$

$\Gamma 331\ -r\mathrm{Sin}[\theta]^2$

$\Gamma 332\ -r^2\mathrm{Cos}[\theta]\mathrm{Sin}[\theta]$

$\Gamma \mathrm{u}010\ \frac{1}{2(-1+r)r}$

$\Gamma \mathrm{u}100\ \frac{-1+r}{2r^3}$

$\Gamma \mathrm{u}111\ \frac{1}{2r-2r^2}$

$\Gamma \mathrm{u}122\ 1-r$

$\Gamma \mathrm{u}133\ -(-1+r)\mathrm{Sin}[\theta]^2$

$\Gamma \mathrm{u}221\ \frac{1}{r}$

$\Gamma \mathrm{u}233\ -\mathrm{Cos}[\theta]\mathrm{Sin}[\theta]$

$\Gamma \mathrm{u}331\ \frac{1}{r}$

$\Gamma \mathrm{u}332\ \mathrm{Cot}[\theta]$

$\mathrm{R}1010\ \frac{1}{r^3}$

$\mathrm{R}2020\ -\frac{-1+r}{2r^2}$

$\mathrm{R}2121\ \frac{1}{2(-1+r)}$

$\mathrm{R}3030\ -\frac{(-1+r)\mathrm{Sin}[\theta]^2}{2r^2}$

$\mathrm{R}3131\ \frac{\mathrm{Sin}[\theta]^2}{-2+2r}$

$\mathrm{R}3232\ -r\mathrm{Sin}[\theta]^2$

The Ricci tensor (and hence the scalar curvature) vanishes. Therefore, the Schwarzschild metric satisfies the vacuum Einstein equation.